

TAIVAANMEKANIIKAN KOTITEHTÄVÄT

(syksy 2008)

1. Marsin rata taivaalla vuosina 2000-2020

2. Kahden kappaleen suhteellisen liikkeen numeerinen integrointi

Tee kummastakin tehtävästä lyhyt selostus= muutama kuva joissa on selvästi kerrottu mitä ne esittävät + hyvin dokumentoitu ohjelmalistaus

Käytä hyväksi IDL-harjoitusmateriaalia, jolloin tehtävät ovat erittäin helpot ratkaista!

HUOM: tehtävien tarkoituksena ei ole viivästyttää kurssin suorittamista! Eli neuvoja saa tulla aina kysymään (TÄ 314, heikki.salo@oulu.fi).

| |
|--|
| TEHTÄVÄ 1: Marsin rata taivaalla vuosina 2000-2020 |
|--|

Laskuharjoituksissa (Harjoitus 1, tehtävä 12) on ollut esimerkkinä Jupiterin paikan laskeminen 1.1.2000 rataelementeistä lähtien (sama esimerkki Karttusen kirjassa).

Tämän ohjeen liitteenä on esimerkki `tm2008_harj1_12.pro` tehtävän ratkaisemisesta käyttäen hyväksi `TM2008_DEMO2.dir` hakemistossa olevaa `elem_to_rv.pro` ohjelmaa, joka laskee planeetan heliosentriset ekliptika-koordinaatit annetusta rataelementeistä.

Liitteenä oleva toinen esimerkki `tm2008_harj1_12_rata.pro` puolestaan laskee Jupiterin paikan 0.01 vuoden välein vuosille 2000-2020. Rataelementtien muutokset on otettu huomioon käyttämällä ”Tähtitieteen Perusteet” kirjan taulukkoa D.12.






Tulokset on tarkastettu vertaamalla niitä IDL:n ASTRO-kirjaston `planet_coords.pro` ohjelman laskemiin. Tama ohjelma laskee paikat kahdella valinnaisella tarkkuudella:

- 1) käyttäen samantapaista approksimaatioita kuin `tm2008_harj1_12_rata.pro`
- 2) käyttäen tarkkoja JPL-ephemeridejä (vaatii ”JPLEPH.405” tiedoston)

Seuraavan sivun kuvissa on esitetty `tm2008_harj1_12_rata.pro` ohjelman tuottamat kuvat.

TEHTÄVÄ: Laske vastaavalla tavalla Marsin sijainti taivaalla v. 2000-2020.

Ohjelman tm2008_harj1_12_rata.pro tulostuksia:



TEHTÄVÄ 2: Suhteellisen liikkeen numeerinen integrointi

1) Tee ohjelma, joka laskee analyyttisen radan ajanhetkille $t = t_0 + i \times \Delta t$, $i = 0, 1, 2, \dots$, valituista rataelementeistä lähtien:

- Valitse rataelementit: a, e, t_0 , ja valitse yksiköt asettamalla $\mu = G(m_1 + m_2) = 1$
- Rataperiodi $P = 2\pi\sqrt{a^3/\mu}$ ($\mu = 1 \Rightarrow$ kierrosaika 2π yksikköetäisyydellä)
- Ratkaise Keplerin yhtälö hetkellä t :

$$M = 2\pi \frac{t-t_0}{P}$$

$$M = E - e \sin E$$

- Paikka ja nopeus radan parametriyhtälöstä ($b = a\sqrt{1-e^2}$)

$$x = a(\cos E - e)$$

$$y = b \sin E$$

$$v_x = -a \sin E \sqrt{\mu/a^3} (1 - e \cos E)^{-1}$$

$$v_y = b \cos E \sqrt{\mu/a^3} (1 - e \cos E)^{-1}$$

$$\text{Hetkellä } t = t_0 \Rightarrow M = 0, E = 0 \Rightarrow [x_0, y_0], [v_{x0}, v_{y0}]$$

2) Integroi sama rata numeerisesti, lähtien valitulle ajanhetkelle lasketuista paikka- ja nopeusvektoreista. Vertaa tuloksia analyyttiseen rataan.

3) Tarkista integroinnin tarkkuus seuraamalla energian ja impulssimomentin säilymistä

$$H = 0.5v^2 - \mu/r$$

$$L_z = (\vec{R} \times \vec{V})_z = xv_y - yv_x$$

Hyödyllistä katsoa mikä on suhteellinen muutos/ratakierros

4) Vertaa seuraavia tapauksia (energian muutoksen perusteella)

- I & II asteen Taylor & RK4 menetelmät
- Vertaa eksentrisyyden arvoja 0.0, 0.5 ja 0.9
- aika-askeleen vaikutus: esim $\Delta t = 0.1, 0.01, 0.001, 0.0001\dots$ rataperiodia, integroinnin kokonaisaika esim. 10 rataperiodia

tm2008_harj1_12.pro

```
;-----
program='tm2008_harj1_12'
;-----

;Taivaanmekaniikan laskuharjoitus 1, tehtava 12:
;Jupiterin paikka taivaalla 1.1.2000

;kaytetaan elem_to_rv ohjelmaa apuna: laskee rataelementit ->
;heliosentrinen paikka ekliptika-systeemissa

;lasketaan Jupiterin ja Maan paikkojen erotus, muunnetaan
;ekvaattori-systeemiin --> alpha,dec

;-----
;Jupiterin rataelementit
;(luennolla jaettu kopio tÄhtitieteen perusteet kirjasta)
;-----

a=5.20336301d0
eks=0.04839266d0
ink=1.30530d0
ome=100.55616d0
wp=14.75385d0      ;perihelin pituus (solmuviivasta)
L=34.40438d0       ;keskipituus (solmuviivasta)

;lasketaan naista:
w=wp-ome           ;perisentrin argumenttti
M=L-wp            ;keskianomalia

;KÄytetään suoraan keskianomaliaa
tau=0.             ;dummy, silla kaytetaan suoraan keskianomaliaa
time=0.            ; "-"

;HUOM katso elem_to_rv ohjelman helppi!

elem=[a,eks,ink,ome,w,tau]
elem_to_rv,elem,time,rad,vel,m0=M!/radeg

rad_jup=rad        ;jupiterin paikka vektori

;-----
;Maan rataelementit
;-----

a=1.00000011d0
eks=0.01671022d0
ink=0.00005d0
ome=-11.26064d0
wp=102.94719d0
L=100.46435d0

w=wp-ome           ;perisentrin argumenttti
M=L-wp            ;keskianomalia

tau=0.             ;dummy, silla kaytetaan suoraan keskianomaliaa
time=0.            ; "-"

elem=[a,eks,ink,ome,w,tau]
elem_to_rv,elem,time,rad,vel,m0=M!/radeg

rad_maa=rad

;erotus: ekliptika-systeemi
x=rad_jup(0)-rad_maa(0)
y=rad_jup(1)-rad_maa(1)
z=rad_jup(2)-rad_maa(2)
```

```

;erotus: ekvaattori-systeemi
ekli=23.4393
sine=sin(ekli/!radeg)
cose=cos(ekli/!radeg)

xe=x
ye=y*cose-z*sine
ze=y*sine+z*cose

re=sqrt(xe^2+ye^2+ze^2)
delta=asin(ze/re)*!radeg
alpha=atan(ye,xe)*!radeg

print,'Jupiter 1.1.2000:'
print,'etaisyys',re
print,'deklinatio',delta
print,'rektaskensio',alpha

;astro-kirjaston ADSTRING muunnos
;asteet -> tunnint,minuutit sekunnit
print,adstring(delta)
print,adstring(alpha)

end

```

tm2008_harj1_12_rata.pro

```
program='tm2008_harj1_12_rata'
ps=0

;jatkoa ohjelmalle tm2008_harj1_12.pro
;lasketaan 1.1.2000 sijainnin sijasta Jupiterin
;naennainen rata vuosina 2000-2020

;Kaytetaan Tahtitieteen perusteet Taulukon D.12 rataelementteja
;eli v. 2000.0 rataelementteihin lisataan aikaan verrannolliset
;korjaukset
;huom: tehdään kaikki laskenta kaksoistarkkuudella

;-----
;lasketaan rata 2000-2020 (21 vuoden ajalle)

timet=dindgen(2100)/100.      ; aika vuosina 21 vuotta, 0.01 vuoden valein
tday=timet*365.25d0          ; aika vuorokausina 1.1.2000 lahtien
tcen=tday/36525.d0           ; aika vuosisatoina

;tallennetaan taulukoihin

;declinaatio, rektaskensio
deltat=timet
alphan=timet

;latitudi, longitudi ekliptika systeemissa
lattab=timet
lontab=timet

for i=01,n_elements(timet)-1 do begin

    T=tcen(i)                  ;aika juliaanisina vuosisatoina

;-----
;Jupiterin rataelementit : lisataan korjaukset (HUOM yksikot!)
;-----

    a   = 5.20336301      + 0.00060737*T
    eks = 0.04839266      - 0.00012880*T
    ink = 1.30530         - 4.15/3600.*T
    ome =100.55616        + 1217.17/3600.*T
    wp  = 14.75385        + 839.93/3600.*T ;perihelin pituus (solmuviivasta)
    L   = 34.40438        + 0.08308676*tday(i) ;keskipituus (solmuviivasta)

    w=wp-ome                ;perisentrin argumenttti
    M=L-wp                  ;keskianomalia

    tau=0.d0 ;dummy, silla annataan M suoraan
    elem_jup=[a,eks,ink,ome,w,tau]
    M_jup=(M mod 360.d0)/360.d0*2.*!dpi

;-----
;Maan rataelementit
;-----

    a   = 1.00000011      - 0.00000005*T
    eks = 0.0167102       - 0.00003804*T
    ink = 0.00005         - 46.94/3600.*T
    ome =-11.26064        - 18228.25/3600.*T
    wp  = 102.94719       + 1198.28/3600*T
    L   = 100.46435       + 0.98560910*tday(i)

    w=wp-ome                ;perisentrin argumenttti
    M=L-wp                  ;keskianomalia

    elem_maa=[a,eks,ink,ome,w,tau]
    M_maa=(M mod 360.d0)/360.d0*2.*!dpi
```

```

time=0.d0 ;dummy koska kaytetaan M
elem_to_rv,elem_jup,time,rad,vel,M0=M_JUP
rad_jup=rad

elem_to_rv,elem_maa,time,rad,vel,M0=M_maa
rad_maa=rad

;erotus: ekliptika-systeemi
x=rad_jup(0)-rad_maa(0)
y=rad_jup(1)-rad_maa(1)
z=rad_jup(2)-rad_maa(2)
r=sqrt(x^2+y^2+z^2)

lattab(i)=asin(z/r)*!radeg
lontab(i)=atan(y,x)*!radeg

;erotus: ekvaattori-systeemi
;kierretaan ekliptikan kaltevuuden verran

ekli=23.43928d0
sine=sin(ekli/!radeg)
cose=cos(ekli/!radeg)

xe=x
ye=y*cose-z*sine
ze=y*sine+z*cose

re=sqrt(xe^2+ye^2+ze^2)
delta=asin(ze/re)*!radeg
alpha=atan(ye,xe)*!radeg

if(alpha le 0) then alpha=alpha+360.

deltat(i)=delta
alphan(i)=alpha

endfor

;-----
;piirretaan rata ekliptika-systeemissa

psdirect,program+'_a',ps,/color

nwin
plot,lontab,lattab,xtitle='longitudi pitkin elliptikaa',
title='Jupiterin rata 2000 -2020', ytitle='latitudi',psym=3

psdirect,program+'_a',ps,/color,/stop

;-----
;tarkistuksen vuoksi lasketaan kayttaen
;ASTRO-kirjaston planet_coords proseduuria

;juldate palauttaa redusoidun Julian Date
; = JD-2400000.0

;alkuhetkelle
juldate,[2000.,1.,1],jd0
jd0=jd0+2400000.d0

jd=jd0+tday ;ylla maaritellyt ajanhetket -> JD

planet_coords,jd,/jd,ra_astro,dec_astro,planet='jupiter'

;tarkempi, kayttaen JPL ephemerideja
planet_coords,jd,/jd,ra_jpl,dec_jpl,planet='jupiter',/jpl

```



```

;-----
psdirect,program+'_b',ps,/color
nwin
!p.multi=[0,2,2]
!p.charsize=0.7

plot,2000+timet,deltat,xr=[0,20]+2000,$
    xtitle='Vuosi',ytitle='Jupiterin deklinaatio',psym=3

;plotaan vain joka 20 piste
index=lindgen(n_elements(timet)/20)*20

oplot,2000+timet(index),dec_astro(index),col=2,psym=6,syms=0.5
oplot,2000+timet(index),dec_astro(index),col=3,psym=1,syms=0.5

plot,2000+timet,alphan,xr=[0,20]+2000,$
    xtitle='Vuosi',ytitle='Jupiterin rektaskensio',psym=3

oplot,2000+timet(index),ra_astro(index),col=2,psym=6,syms=0.5
oplot,2000+timet(index),ra_astro(index),col=3,psym=1,syms=0.5

plot,2000+timet,(deltat-dec_jpl)*60.,xr=[0,20]+2000,$
    xtitle='Vuosi',ytitle='deklinaation virhe (arcmin)',psym=3

;alkuperaiset alpha ja ra_jpl voivat poiketa 360 verran
d_alpha=atan(tan((alphan-ra_jpl)/!radeg))*!radeg

plot,2000+timet,d_alpha*60,xr=[0,20]+2000,$
    xtitle='Vuosi',ytitle='rektaskension virhe (arcmin)',psym=3

!p.charsize=1
!p.multi=0
psdirect,program+'_b',ps,/color,/stop

end

```

TÄHTITIEDEEN PERUSTEET TAULUKKO D.12

Taulukoita

595

Taulukko D.12. Planeettojen rataelementit epookin J2000.0 ekvaattorin ja ta-
sauspisteen suhteen. Seuraavissa sarjoissa t on aika vuorokausina epookista J2000.0
ja T sama aika juliaanisin vuosisatoina: $t = J - 2451545.0$, $T = t/36525$. L on kes-
kilongitudi, $L = M + \varpi$. Elementit ovat vain keskimääräisiä, ja niistä laskettujen he-
liosentristen paikkojen tarkkuus on muutamia kaariminuutteja. Luvut ovat teoksesta
Explanatory Supplement to the Astronomical Almanac. Maan elementit kuvaavat
Maan ja Kuun muodostaman järjestelmän painopisteen rataa.

| | | |
|-----------|---|--|
| Merkurius | $a = 0.38709893 + 0.00000066 T$ $i = 7.00487^\circ - 23.51'' T$ $\varpi = 77.45645^\circ + 573.57'' T$ | $e = 0.20563069 + 0.00002527 T$ $\Omega = 48.33167^\circ - 446.30'' T$ $L = 252.25084^\circ + 4.09233880^\circ t$ |
| Venus | $a = 0.72333199 + 0.00000092 T$ $i = 3.39471^\circ - 2.86'' T$ $\varpi = 131.53298^\circ - 108.80'' T$ | $e = 0.00677323 - 0.00004938 T$ $\Omega = 76.68069^\circ - 996.89'' T$ $L = 181.97973^\circ + 1.60213047^\circ t$ |
| Maa+Kuu | $a = 1.00000011 - 0.00000005 T$ $i = 0.00005^\circ - 46.94'' T$ $\varpi = 102.94719^\circ + 1198.28'' T$ | $e = 0.01671022 - 0.00003804 T$ $\Omega = -11.26064^\circ - 18228.25'' T$ $L = 100.46435^\circ + 0.98560910^\circ t$ |
| Mars | $a = 1.52366231 - 0.00007221 T$ $i = 1.85061^\circ - 25.47'' T$ $\varpi = 336.04084^\circ + 1560.78'' T$ | $e = 0.09341233 + 0.00011902 T$ $\Omega = 49.57854^\circ - 1020.19'' T$ $L = 355.45332^\circ + 0.52403304^\circ t$ |
| Jupiter | $a = 5.20336301 + 0.00060737 T$ $i = 1.30530^\circ - 4.15'' T$ $\varpi = 14.75385^\circ + 839.93'' T$ | $e = 0.04839266 - 0.00012880 T$ $\Omega = 100.55615^\circ + 1217.17'' T$ $L = 34.40438^\circ + 0.08308676^\circ t$ |
| Saturnus | $a = 9.53707032 - 0.00301530 T$ $i = 2.48446^\circ + 6.11'' T$ $\varpi = 92.43194^\circ - 1948.89'' T$ | $e = 0.05415060 - 0.00036762 T$ $\Omega = 113.71504^\circ - 1591.05'' T$ $L = 49.94432^\circ + 0.03346063^\circ t$ |
| Uranus | $a = 19.19126393 + 0.00152025 T$ $i = 0.76986^\circ - 2.09'' T$ $\varpi = 170.96424^\circ + 1312.56'' T$ | $e = 0.04716771 - 0.00019150 T$ $\Omega = 74.22988^\circ + 1681.40'' T$ $L = 313.23218^\circ + 0.01173129^\circ t$ |
| Neptunus | $a = 30.06896348 - 0.00125196 T$ $i = 1.76917^\circ - 3.64'' T$ $\varpi = 44.97135^\circ - 844.43'' T$ | $e = 0.00858587 + 0.00002514 T$ $\Omega = 131.72169^\circ - 151.25'' T$ $L = 304.88003^\circ + 0.00598106^\circ t$ |
| Pluto | $a = 39.48168677 - 0.00076912 T$ $i = 17.14175^\circ + 11.07'' T$ $\varpi = 224.06676^\circ - 132.25'' T$ | $e = 0.24880766 + 0.00006465 T$ $\Omega = 110.30347^\circ - 37.33'' T$ $L = 238.92881^\circ + 0.00397557^\circ t$ |